

*)9BEG

% =====
% = ROUTINE 11.1 =
% =====

S3RT-FL0DU-NPL:54MB

%
% PROGRAM TO LAY OUT A SINTRAN III RT SYSTEM
% ON A DISKETTE
%
% VALID FOR: S-III 80.02.01.A
%
% MODIFIED: 80.03.21 (LA)
%

*)9ENT S3RDU

BASE AA
INTEGER FORM,CUNIT,CHCB,SAVX,SAVT,BUFFP:=BUFFER
INTEGER DESTFILE,SOURCEFILE
INTEGER POINTER OUTTEXT:=9OUTTEXT,READTEXT:=8READTEXT,INBT:=9INBT
INTEGER POINTER OUTBT:=9OUTBT,MOVCHAR:=9MOVCHAR
INTEGER POINTER STSPACE:=9STSPACE,STAPO:=9STAPO,STCHAR:=9STCHAR
INTEGER POINTER MON70:=9MON70,COPYTEXT:=TEXTCOPY,COPYFILE:=FILECOPY
INTEGER PCINTER LINK1,LINK2
ESAB

INTEGER ADIRNAM:='\$DIRECTORY NAME: '
INTEGER AUNITNO:='\$DEVICE UNIT: '
INTEGER ADEVNAM:='\$DEVICE NAME: '
INTEGER AFORMAT:='\$FORMATING THE DISKETTE? '
INTEGER APERIFILE:='\$PERIPHERAL FILE NAME: '
INTEGER CREDIR:='CREATE-DIRECTORY'
INTEGER ENTDIR:='ENTER-DIRECTORY'
INTEGER CREUSER:='CREATE-USER'
INTEGER GIVUSER:='GIVE-USER-SPACE'
INTEGER ALLFILE:='ALLOCATE-FILE'
INTEGER DEVFU:='DEVICE-FUNCTION'
INTEGER FORMAT:='FORMAT'
INTEGER RELDIR:='RELEASE-DIRECTORY'
INTEGER SYSTEM:='SYSTEM'
INTEGER BITFADR:='231B'

```
INTEGER NOPAGES:='148'  
INTEGER BOOTSTRAP:=' DUMP-BOOT FLOPPY-MONITOR'  
INTEGER PTERMINATE:='$--- EXIT ---$'  
INTEGER FILCOPY:='COPY-FILE ('  
INTEGER SYMBLI:=':SYSTEM)SYMBOL-LIST:SYMB 1 21B'  
INTEGER SINTRAN:=':SYSTEM)SINTRAN-III-RT:BPUN 229 101B'  
INTEGER BACKUP:=':SYSTEM)BACKUP-III-RT:BPUN 123B 101B'  
INTEGER BCOP1:=':SYSTEM)SINTRAN-III-RT:BPUN'  
INTEGER BCOP2:=':SYSTEM)BACKUP-III-RT:BPUN'  
INTEGER BCOP3:=':SYSTEM)SYMBOL-LIST:SYMB'  
INTEGER BPUFI:=' (AUX-SIN)SI3RT-00:BPUN'  
INTEGER SLIST:=' (AUX-SIN)RTLIS-00:SYMB'  
INTEGER ARRAY DIRNAM(40),PERIFILE(40),DEVNAM(40),BUFFER(140),YESNO(40)  
INTEGER SAVB,SAVL  
DISP 0; INTEGER POINTER JOBNO; PSID % PARAMETER POINTER  
DISP 10; INTEGER FINO; PSID % DISPLACEMENT IN FILE-NAMES
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%  
%% JOB NUMBERS IN ASCII. (00-99)  
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%  
@ICR;
```

```
INTEGER ARRAY JBNO:=(  
30060,30061,30062,30063,30064,30065,30066,30067,30070,30071,  
30460,30461,30462,30463,30464,30465,30466,30467,30470,30471,  
31060,31061,31062,31063,31064,31065,31066,31067,31070,31071,  
31460,31461,31462,31463,31464,31465,31466,31467,31470,31471,  
32060,32061,32062,32063,32064,32065,32066,32067,32070,32071,  
32460,32461,32462,32463,32464,32465,32466,32467,32470,32471,  
33060,33061,33062,33063,33064,33065,33066,33067,33070,33071,  
33460,33461,33462,33463,33464,33465,33466,33467,33470,33471,  
34060,34061,34062,34063,34064,34065,34066,34067,34070,34071,  
34460,34461,34462,34463,34464,34465,34466,34467,34470,34471);  
@OCR;
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%  
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%  
%% ENTRYPOINT CALLED FROM FORTRAN.
```

```
SUBR S3RDU  
S3RDU: A:=B:=SAVB  
A:=L:=SAVL  
  
X:=JOBNO; A:=JBNO(X) % GET PARAMETER AND ASCII EQUIV.  
X:="BPUFI"; A:=X.FINO % MODIFY FILENAME WITH JOB NO  
X:="SLIST"; A:=X.FINO  
  
CALL DUMP  
SAVB=:B  
SAVL=:L  
EXIT
```

```
RBUS  
%% RETURN TO FORTRAN.  
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%  
SUBR DUMP  
DUMP: "AA"=:B  
A:=L:="LINK2"  
O:=FORM=:CUNIT
```

```
% ASK IF DISKETTE SHOULD BE FORMATED  
ASK1: "AFORMAT"; CALL OUTTEXT; T:="YESNO"; CALL READTEXT  
X:=0; "YESNO"=:D; *LBYT; AAX 1  
IF A=##Y THEN  
T:=D; *LBYT; AAX 1  
IF A><##E AND A><##' GO ASK1  
IF A=##' GO YES  
T:=D; *LBYT; AAX 1  
IF A ><##S AND A><##' GO ASK1  
IF A=##' GO YES  
T:=D; *LBYT; AAX 1  
IF A><##' GO ASK1  
MIN FORM  
YES:  
ELSE  
IF A=##N THEN  
T:=D; *LBYT; AAX 1  
IF A><##O AND A><##' GO ASK1  
IF A=##' GO LX  
T:=D; *LBYT  
IF A><##' GO ASK1  
ELSE  
GO ASK1  
FI; FI
```

```
% READ THE PERIPHERAL FILE NAME OF THE FLOPPY DISC  
LX: "APERIFILE"; CALL OUTTEXT; T:="PERIFILE"; CALL READTEXT  
  
% READ DIRECTORY NAME  
"ADIRNAM"; CALL OUTTEXT; T:="DIRNAM"; CALL READTEXT  
  
% READ DEVICE NAME  
"ADEVNAM"; CALL OUTTEXT; T:="DEVNAM"; CALL READTEXT  
  
% READ UNIT NUMBER
```

```

ASKUNIT: "AUNITNO"; CALL OUTTEXT; CALL INBT
IF A<#0 OR A>##2 AND A><##N GO ASKUNIT
A=:CUNIT; DO CALL INBT WHILE A><15 OD
IF FORM><0 THEN
% FORMAT THE DISKETTE
O=:CHCB; T:="DEVFU"; CALL MOVCHAR; CALL STSPACE
T:="PERIFI"; CALL MOVCHAR; CALL STSPACE
T:="FORMAT"; CALL MOVCHAR; CALL STAPO
CALL MON70; GO L2; *)FILL
FI
L2: O=:CHCB
% CREATE DIRECTORY
T:="CREDIR"; CALL MOVCHAR; CALL STSPACE
T:="DIRNAM"; CALL MOVCHAR; CALL STSPACE
T:="DEVNAM"; CALL MOVCHAR
IF CUNIT><##N THEN
CALL STSPACE; CUNIT; CALL STCHAR
FI; CALL STSPACE; T:="BITFADR"; CALL MOVCHAR
CALL STAPO; CALL MON70
% ENTER DIRECTORY
O=:CHCB
T:="ENTDIR"; CALL MOVCHAR; CALL STSPACE
T:="DIRNAM"; CALL MOVCHAR; CALL STSPACE
T:="DEVNAM"; CALL MOVCHAR
IF CUNIT><##N THEN
CALL STSPACE; CUNIT; CALL STCHAR
FI; CALL STAPO; CALL MON70
% CREATE USER
O=:CHCB
T:="CREUSER"; CALL MOVCHAR; CALL STSPACE
T:="DIRNAM"; CALL MOVCHAR; ##:; CALL STCHAR
T:="SYSTEM"; CALL MOVCHAR; CALL STAPO
CALL MON70; GO L3; *)FILL
% GIVE USER SPACE
L3: O=:CHCB
T:="GIVUSER"; CALL MOVCHAR; CALL STSPACE
T:="DIRNAM"; CALL MOVCHAR; ##:; CALL STCHAR
T:="SYSTEM"; CALL MOVCHAR; CALL STSPACE
T:="NOPAGES"; CALL MOVCHAR; CALL STAPO
CALL MON70
% ALLOCATE THE SYMBOL-LIST FILE
O=:CHCB
T:="ALLFILE"; CALL MOVCHAR; CALL STSPACE
##(; CALL STCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="SYMLI"; CALL MOVCHAR; CALL STAPO; CALL MON70
% ALLOCATE THE SINTRAN FILE
O=:CHCB
T:="ALLFILE"; CALL MOVCHAR; CALL STSPACE
##(; CALL STCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="SINTRAN"; CALL MOVCHAR; CALL STAPO; CALL MON70
% ALLOCATE THE BACKUP FILE
O=:CHCB
T:="ALLFILE"; CALL MOVCHAR; CALL STSPACE
##(; CALL STCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="BACKUP"; CALL MOVCHAR; CALL STAPO; CALL MON70
% COPY THE BPUNFILE TO THE SINTRAN-FILE
O=:CHCB
T:="FILCOPY"; CALL MOVCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="BCOP1"; CALL MOVCHAR; T:="BPUFI"; CALL MOVCHAR
CALL STAPO; CALL MON70; GO L4; *)FILL
% COPY THE BPUNFILE TO THE BACKUP-FILE
L4: O=:CHCB
T:="FILCOPY"; CALL MOVCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="BCOP2"; CALL MOVCHAR; T:="BPUFI"; CALL MOVCHAR
CALL STAPO; CALL MON70
% COPY THE SYMBOL-LIST TO THE SYMBOL-LIST FILE
O=:CHCB
T:="FILCOPY"; CALL MOVCHAR; T:="DIRNAM"; CALL MOVCHAR
T:="BCOP3"; CALL MOVCHAR; T:="SLIST"; CALL MOVCHAR
CALL STAPO; CALL MON70
% CLOSE ALL FILES
T:=-1; *MON 43; JMP *+1
% RELEASE DIRECTORY
O=:CHCB
T:="RELDIR"; CALL MOVCHAR; CALL STSPACE
T:="DIRNAM"; CALL MOVCHAR; CALL STAPO; CALL MON70
% DUMP BOOTSTRAP
O=:CHCB
T:="DEVFU"; CALL MOVCHAR; CALL STSPACE
T:="PERIFI"; CALL MOVCHAR; T:="BOOTSTRAP"; CALL MOVCHAR
CALL STAPO; CALL MON70

```

```

% PROGRAM IS TERMINATED
"PTERM"; CALL OUTTEXT
GO LINK2
RBUS

%
% READTEXT
% SUBROUTINE TO READ A TEXT STRING FROM THE TERMINAL
% THE INPUT SHOULD BE TERMINATED WITH CARRIAGE RETURN
%
% ENTRY: T=ADDRESS OF THE ARRAY WHERE THE TEXT STRING SHOULD
% BE STORED
%
% RETURN: TEXT STRING TERMINATED WITH THE CHARACTER "'" IN THE ARRAY
%

SUBR 8READTEXT
8READTEXT: A:=L:="LINK1"
X:=0; T:=D
DO CALL INBT WHILE A><15
IF A=21 THEN ##_ ; CALL OUTBT; X:=0; GO NEXT FI
IF A=1 THEN ##^ ; CALL OUTBT; X-1; GO NEXT FI
IF X>100 GO NEXT
T:=D; *SBYT; AAX 1
NEXT: OD; T:=D; ##' ; *SBYT
GO LINK1
RBUS

%
% INBT
% SUBROUTINE TO READ ONE CHARACTER FROM THE TERMINAL
%
% RETURN: A=CHARACTER
%

SUBR 9INBT
9INBT: T:=SAVT:=1; *MON 1; MON 65
A/0177; T:=SAVT; EXIT
RBUS

%
% MOVCHAR
% SUBROUTINE TO MOVE CHARACTERS FROM A TEXT STRING TO AN ARRAY
%
% ENTRY: T=ADDRESS OF TEXT STRING
%

SUBR 9MOVCHAR
9MOVCHAR: A:=L:="LINK1"
T:=D; X:=0
DO T:=D; *LBYT
WHILE A><##' ; CALL STCHAR; X+1
OD; GO LINK1
RBUS

%
% STCHAR
% SUBROUTINE TO STORE ONE CHARACTER IN THE ARRAY NAMED BUFFER
%
% ENTRY: A=CHARACTER
%

SUBR 9STCHAR
9STCHAR: X:=SAVX:=CHCB; T:=BUFFP; *SBYT
MIN CHCB; X:=SAVX; EXIT
RBUS

%
% STAPO - STSPACE
% SUBROUTINES TO STORE A SPACE OR A "'" IN THE ARRAY NAMED BUFFER
%

SUBR 9STAPO,9STSPACE
9STAPO: ##' ; GO STCHAR
9STSPACE: 40; GO STCHAR
RBUS

%
% OUTTEXT
% SUBROUTINE TO PRINT A TEXT STRING ON THE TERMINAL
%
% ENTRY: A=ADDRESS OF THE TEXT STRING
%

SUBR 9OUTTEXT
9OUTTEXT: T:=L:="LINK1"
A:=D; X:=0
DO T:=D; *LBYT
WHILE A><##'
IF A=##$ THEN 15; CALL OUTBT; 12 FI
CALL OUTBT; X+1
OD; GO LINK1
RBUS

```

```
%  
% OUTBT  
% SUBROUTINE TO OUTPUT ONE CHARACTER ON THE TERMINAL  
%  
% ENTRY: A=CHARACTER  
%
```

```
SUBR 9OUTBT  
9OUTBT: T:=SAVT:=1; *MON 2; MON 65  
EXIT
```

```
RBUS
```

```
%  
% MON70  
% SUBROUTINE TO EXECUTE THE MONITOR CALL COMND (MON 70)  
%
```

```
SUBR 9MON70  
9MON70: A:=BUFFP; *MON 70; EXIT  
RBUS
```

```
%  
% COPYTEXT  
% SUBROUTINE TO COPY A TEXT STRING TO THE DISKETTE  
%  
% ENTRY: A=ADDRESS OF TEXT STRING  
%
```

```
SUBR TEXTCOPY  
TEXTCOPY: A=:D; X:=0  
DO  
T:=D; *LBYT; AAX 1  
WHILE A><##'  
* BSET ZRO 70 DA; BSKP ZRO 60 DA; BSET BCM 70 DA  
* BSKP ZRO 50 DA; BSET BCM 70 DA; BSKP ZRO 40 DA  
* BSET BCM 70 DA; BSKP ZRO 30 DA; BSET BCM 70 DA  
* BSKP ZRO 20 DA; BSET BCM 70 DA; BSKP ZRO 10 DA  
* BSET BCM 70 DA; BSKP ZRO 00 DA; BSET BCM 70 DA  
T:=DESTFILE; *MON 2; MON 65  
OD; EXIT
```

```
RBUS
```

```
%  
% COPYFILE  
% SUBROUTINE TO COPY A FILE TO THE DISKETTE  
%
```

```
SUBR FILECOPY  
FILECOPY: DO  
T:=SOURCEFILE; *MON 1; JMP OUT  
T:=DESTFILE; *MON 2; MON 65  
OD  
OUT: IF A=3 THEN EXIT FI  
*MON 65
```

```
RBUS
```

```
@MAC  
)9END  
)9EOF  
@  
@EOF
```